-- Non-blocking circular bounded buffers

generic
type Element is private;
package NXT.Buffers is
  pragma Elaborate_Body;

  type Buffer (Capacity : Positive) is limited private;

  procedure Insert (Item : Element; Into : in out Buffer);
  -- Insert Item into the buffer, overwriting if necessary.
  -- Does not block caller if no space is available.

  procedure Remove (Item : out Element; From : in out Buffer);
  -- Remove next available Element from buffer.
  -- Does not block caller is none available.

  function Empty (This : Buffer) return Boolean;
  -- Returns whether the instance contains any Elements.

  function Full (This : Buffer) return Boolean;
  -- Returns whether any space remains within the instance.

  function Extent (This : Buffer) return Natural;
  -- Returns the number of Element values currently held
  -- within the instance.

  procedure Clear (This : in out Buffer);
  -- Makes This buffer become empty

  generic
    with procedure Visit (This : Element);
  procedure Iteration (Over : Buffer);
private

    type Content is array (Positive range <> ) of Element;

    type Buffer (Capacity : Positive) is
        record
            Values  : Content (1 .. Capacity);
            Next_In : Positive := 1;
            Next_Out : Positive := 1;
            Count   : Natural := 0;
        end record;

end NXT.Buffers;